


# CALLNAT



This chapter covers the following topics:

- Function
- Syntax Description
- Parameter Transfer with Dynamic Variables
- Examples

For an explanation of the symbols used in the syntax diagram, see *Syntax Symbols*.

Related Statements: `CALL` | `CALL FILE` | `CALL LOOP` | `DEFINE SUBROUTINE` | `ESCAPE` | `FETCH` | `PERFORM`

Belongs to Function Group: *Invoking Programs and Routines*

## Function

The `CALLNAT` statement is used to invoke a Natural subprogram for execution. (A Natural subprogram can only be invoked via a `CALLNAT` statement; it cannot be executed by itself.)

When the `CALLNAT` statement is executed, the execution of the invoking object (that is, the object containing the `CALLNAT` statement) will be suspended and the invoked subprogram will be executed. The execution of the subprogram continues until either its `END` statement is reached or processing of the subprogram is stopped by an `ESCAPE ROUTINE` statement being executed. In either case, processing of the invoking object will then continue with the statement following the `CALLNAT` statement.

### Notes:

1. A subprogram can in turn invoke other subprograms.
2. A subprogram has no access to the global data area used by the invoking object. If a subprogram in turn invokes a subroutine or helproutine, it can establish its own global data area to be shared with the subroutine/helproutine.

# Syntax Description

Operand Definition Table:

Operand	Possible Structure				Possible Formats												Referencing Permitted	Dynamic Definition	
<i>operand1</i>	C	S			A													yes	no
<i>operand2</i>	C	S	A	G	A	U	N	P	I	F	B	D	T	L	C	G	O	yes	yes

Syntax Element Description:

Syntax Element	Description
<i>operand1</i>	<p><b>Subprogram to be Invoked:</b></p> <p>As <i>operand1</i>, you specify the name of the subprogram to be invoked. The name may be specified either as a constant of 1 to 8 characters, or - if different subprograms are to be called dependent on program logic - as an alphanumeric variable of length 1 to 8.</p> <p>The subprogram name may contain an ampersand (&amp;); at execution time, this character will be replaced by the one-character code corresponding to the current value of the system variable *LANGUAGE. This makes it possible, for example, to invoke different subprograms for the processing of input, depending on the language in which input is provided.</p>

Syntax Element	Description
<i>operand2</i>	<p><b>Parameters:</b></p> <p>If parameters are passed to the subprogram, the structure of the parameter list must be defined in a <code>DEFINE DATA PARAMETER</code> statement. The parameters specified with the <code>CALLNAT</code> statement are the only data available to the subprogram from the invoking object.</p> <p>By default, the parameters are passed <i>by reference</i>, that is, the data are transferred via address parameters, the parameter values themselves are not moved. However, it is also possible to pass parameters <i>by value</i>, that is, pass the actual parameter values. To do so, you define these fields in the <code>DEFINE DATA PARAMETER</code> statement of the subprogram with the option <code>BY VALUE</code> or <code>BY VALUE RESULT</code> (see <i>parameter-data-definition</i> in the description of the <code>DEFINE DATA</code> statement).</p> <ul style="list-style-type: none"> <li>● If parameters are passed <i>by reference</i>, the following applies: The sequence, format and length of the parameters in the invoking object must match exactly the sequence, format and length of the <code>DEFINE DATA PARAMETER</code> structure in the invoked subprogram. The names of the variables in the invoking object and the invoked subprogram may be different.</li> <li>● If parameters are passed <i>by value</i>, the following applies: The sequence of the parameters in the invoking object must match exactly the sequence in the <code>DEFINE DATA PARAMETER</code> structure of the invoked subprogram. Formats and lengths of the variables in the invoking object and the subprogram may be different; however, they have to be data transfer compatible; see the corresponding table in the section <i>Rules for Arithmetic Assignments, Data Transfer</i> in the <i>Programming Guide</i>. The names of the variables in the invoking object and the subprogram may be different. If parameter values that have been modified in the subprogram are to be passed back to the invoking object, you have to define these fields with <code>BY VALUE RESULT</code>. When <code>BY VALUE</code> is specified without <code>RESULT</code>, it is not possible to pass modified parameter values back to the invoking object (regardless of the <code>AD</code> specification; see also below).</li> </ul> <p><b>Note:</b> With <code>BY VALUE</code>, an internal copy of the parameter values is created. The subprogram accesses this copy and can modify it, but this will not affect the original parameter values in the invoking object. With <code>BY VALUE RESULT</code>, an internal copy is likewise created, however, after termination of the subprogram, the original parameter values are overwritten by the (modified) values of the copy.</p> <p>For both ways of passing parameters, the following applies:</p> <p>If a group is specified as <i>operand2</i>, the individual fields contained in that group are passed to the subprogram; that is, for each of these fields a corresponding field must be defined in the subprogram's parameter data area.</p> <p>In the parameter data area of the invoked subprogram, a redefinition of groups is only permitted within a <code>REDEFINE</code> block.</p> <p>If an array is passed, its number of dimensions and occurrences in the subprogram's parameter data area must be the same as in the <code>CALLNAT</code> parameter list.</p> <p><b>Note:</b> If multiple occurrences of an array that is defined as part of an indexed group are passed with the <code>CALLNAT</code> statement, the corresponding fields in the subprogram's parameter data area must not be redefined, as this would lead to the wrong addresses being passed.</p> <p>When the option <code>PCHECK</code> of the <code>COMPOPT</code> command is set to <code>ON</code>, the compiler will check the number, format, length and array index bounds of the parameters that are specified in a <code>CALLNAT</code> statement. Also, the <code>OPTIONAL</code> feature of the <code>DEFINE DATA PARAMETER</code> statement is considered in the parameter check.</p>

Syntax Element	Description
AD=	<b>Attribute Definition:</b> If <i>operand2</i> is a variable, you can mark it in one of the following ways:
	AD=O Non-modifiable, see session parameter AD=O.  <b>Note:</b> Internally, AD=O is processed in the same way as BY VALUE (see <i>parameter-data-definition</i> in the description of the DEFINE DATA statement).
	AD=M Modifiable, see session parameter AD=M.  This is the default setting.
	AD=A Input only, see session parameter AD=A.
	If <i>operand2</i> is a constant, AD cannot be explicitly specified. For constants AD=O always applies.
nX	<b>Parameters to be Skipped:</b>  With the notation <i>nX</i> you can specify that the next <i>n</i> parameters are to be skipped (for example, 1X to skip the next parameter, or 3X to skip the next three parameters); this means that for the next <i>n</i> parameters no values are passed to the subprogram. The possible range of values for <i>n</i> is 1 - 4096.  A parameter that is to be skipped must be defined with the keyword OPTIONAL in the subprogram's DEFINE DATA PARAMETER statement. OPTIONAL means that a value can - but need not - be passed from the invoking object to such a parameter.

## Parameter Transfer with Dynamic Variables

Dynamic variables may be passed as parameters to a called program object (CALLNAT, PERFORM). A call by reference is possible because the value space of a dynamic variable is contiguous. A call by value causes an assignment with the variable definition of the caller as the source operand and the parameter definition as the destination operand. In addition, a call by value result causes the movement to change to the opposite direction. When using a call-by-reference, both definitions must be DYNAMIC. If only one of them is DYNAMIC, a runtime error is raised. In case of a call by value (result) all combinations are possible.

The following table illustrates the valid combinations of statically and dynamically defined variables of the caller, and statically and dynamically defined parameters concerning the parameter transfer.

### Call By Reference

<i>operand2</i> of caller	Parameter definition	
	Static	Dynamic
Static	yes	no
Dynamic	no	yes

The formats of the dynamic variables A or B must match.

## Call by Value (Result)

<i>operand2</i> of caller	Parameter definition	
	Static	Dynamic
Static	yes	yes
Dynamic	yes	yes

### Note:

When using static/dynamic or dynamic/static definitions, a value truncation may occur according to the data transfer rules of the appropriate assignments.

## Examples

- Example 1
- Example 2

### Example 1

#### Calling Program:

```

** Example 'CNTEX1': CALLNAT
*****
DEFINE DATA LOCAL
1 #FIELD1 (N6)
1 #FIELD2 (A20)
1 #FIELD3 (A10)
END-DEFINE
*
CALLNAT 'CNTEX1N' #FIELD1 (AD=M) #FIELD2 (AD=O) #FIELD3 'P4 TEXT'
*
WRITE '=' #FIELD1 '=' #FIELD2 '=' #FIELD3
*
END

```

#### Called Subprogram CNTEX1N:

```

** Example 'CNTEX1N': CALLNAT (called by CNTEX1)
*****
DEFINE DATA PARAMETER
1 #FIELDA (N6)
1 #FIELDB (A20)
1 #FIELD C (A10)
1 #FIELD D (A7)
END-DEFINE
*
*
#FIELDA := 4711
*
#FIELDB := 'HALLO'
*
#FIELD C := 'ABC'

```

```

*
WRITE '=' #FIELDA '=' #FIELDB '=' #FIELDC '=' #FIELDD
*
END

```

## Example 2

### Calling Program:

```

** Example 'CNTEX2': CALLNAT
*****
DEFINE DATA LOCAL
1 #ARRAY1 (N4/1:10,1:10)
1 #NUM (N2)
END-DEFINE
*
*
CALLNAT 'CNTEX2N' #ARRAY1 (2:5,*)
*
FOR #NUM 1 TO 10
    WRITE #NUM #ARRAY1(#NUM,1:10)
END-FOR
*
END

```

### Called Subprogram CNTEX2N:

```

** Example 'CNTEX2N': CALLNAT (called by CNTEX2)
*****
DEFINE DATA
PARAMETER
1 #ARRAY (N4/1:4,1:10)
LOCAL
1 I (I2)
END-DEFINE
*
*
FOR I 1 10
    #ARRAY(1,I) := I
    #ARRAY(2,I) := 100 + I
    #ARRAY(3,I) := 200 + I
    #ARRAY(4,I) := 300 + I
END-FOR
*
END

```